
Abkürzungsverzeichnis für ein Word-Dokument automatisch erstellen:

Akronyme tabellieren

Volker Thormählen, Ratingen

28.06.2016, neue Fassung 02.02.2021

Zur Erstellung eines Abkürzungsverzeichnisses besitzt das Textprogramm MS Word keine Standard-Funktionalität wie bei Abbildungs- und Tabellenverzeichnissen. In diesem Beitrag wird deshalb beschrieben, wie die in einem WORD-Dokument gefundenen Akronyme im entsprechenden Abkürzungsverzeichnis mithilfe von VBA-Prozeduren automatisch tabelliert werden können. Dieser Vorgang kann beliebig oft wiederholt werden, was bei nachträglichen Änderungen eines umfangreichen WORD-Dokuments, die sich auf die Seitennummerierung auswirken, vorteilhaft ist.

Inhaltsverzeichnis

Inhaltsverzeichnis	I
Listingsverzeichnis	II
Abbildungsverzeichnis	III
Tabellenverzeichnis	IV
Abkürzungsverzeichnis	V
1 Aufgabenstellung	1
2 Voraussetzungen	2
3 WORD-Dokumente	3
4 Tabellenaufbau	4
5 Organisation der Prozeduren	5
5.1 Abkürzungsverzeichnis finden	6
5.2 Tabelle einfügen.....	6
5.3 Suchbereich bestimmen	6
5.4 Suchbereich durchlaufen	6
5.5 Suchergebnisse ausgeben.....	7
5.6 Umgebenden Text finden	7
6 Funktionstest	8
7 Zusammenfassung	9
Listings der Prozeduren	V
Literatur-/Quellenverzeichnis	XI

Listingsverzeichnis

Listing 1: Das Wort <i>Abkürzungsverzeichnis</i> im Zieldokument finden	V
Listing 2: Gefundene Akronyme tabellieren, Teil 1 von 3	VI
Listing 3: Gefundene Akronyme tabellieren, Teil 2 von 3	VII
Listing 4: Gefundene Akronyme tabellieren, Teil 3 von 3	VIII
Listing 5: Definition der Akronyme ermitteln bei bestimmter Schreibweise im Text	IX
Listing 6: Alte Tabelle der Akronyme löschen und zugehörige Textmarke erneuern	X
Listing 7: Neue Tabelle der Akronyme formatieren	X

Abbildungsverzeichnis

Abbildung 1: Zieldokument(e) und Quelldokument	3
Abbildung 2: Organisation der Prozeduren.....	5

Tabellenverzeichnis

Tabelle 1: Geordnete Tabelle der Akronyme im WORD-Zieldokument.....V

Abkürzungsverzeichnis¹

Akronym	Definition	Seite
AUDI	(nicht gefunden)	4
BWW	Bayerische Motorenwerke	3
VW	Volkswagen	2

Tabelle 1: Geordnete Tabelle der Akronyme im WORD-Zieldokument

¹ Formatiert mit der benutzerdefinierten Formatvorlage *ÜberschriftOhneNr*

1 Aufgabenstellung

In diesem Beitrag wird beschrieben, wie alle relevanten Akronyme² in einem WORD-Dokument gefunden und als geordnete Tabelle in das dazugehörige Abkürzungsverzeichnis eingefügt werden können. Der Suchvorgang erfolgt durch Prozeduren, die mit der Programmiersprache VBA (*Visual Basic for Applications*) erstellt wurden. Im Folgenden wird im Detail beschrieben,

- welche Voraussetzungen erfüllt sein müssen, um den Suchvorgang nach Akronymen zu automatisieren,
- welche Spalten die Tabelle der Akronyme umfasst,
- welche Prozeduren bzw. Makros³ für das Tabellarisieren der Akronyme notwendig sind,
- wie diese Prozeduren organisiert sind,
- wie sie getestet und eingesetzt werden können.

Der Beitrag endet mit einer kurzen Zusammenfassung und den Listings der eingesetzten Makros bzw. Prozeduren.

² Ein Akronym ist ein Kurzwort, das aus den Anfangsbuchstaben mehrerer Wörter zusammengesetzt ist.

³ Ein Makro ist eine kleine Prozedur, die bestimmte Abläufe in einem Anwendungsprogramm automatisiert.

2 Voraussetzungen

Das WORD-Dokument, das die zu tabellierenden Akronyme enthält (sog. Zieldokument), muss folgende Eigenschaften besitzen:

- Es muss in zwei oder mehr logische Abschnitte (engl. sections) eingeteilt sein, wie z. B. bei wissenschaftlichen Arbeiten üblich⁴.
- Jeder logische Abschnitt muss mit einen *Abschnittsumbruch (Neue Seite)* enden.
- Das Wort „*Abkürzungsverzeichnis*“ muss im Zieldokument nach dem Haupttext vorkommen und mit der benutzerdefinierten Vorlage „*ÜberschriftOhneNr*“⁵ formatiert sein.
- Akronyme müssen im Text des Zieldokuments als zwei oder mehr aufeinander folgende Großbuchstaben vorkommen.

Als weitere wichtige Voraussetzung kommt hinzu, dass das Ziel- und Quelldokument (letzteres enthält die anzuwendenden VBA-Prozeduren) im selben Verzeichnispfad existent sein müssen.

Optional kann die Definition eines Akronyms im Haupttext gemäß folgenden syntaktischen Regeln aufgenommen werden:

- Die Bedeutung steht in runden Klammern **vor** dem Akronym, also: **(Definition) Akronym**
- Die Bedeutung folgt in runden Klammern **nach** dem Akronym, also: **Akronym (Definition)**

Beide Schreibweisen können im selben Zieldokument vorkommen.

⁴ Vgl. dazu die 8-teilige Video-Serie von (Puls, Erika, 2013).

⁵ Vgl. dazu <https://www.fh-bielefeld.de/medienportal/video/Word-Ueberschriften-ohne-Nummerierung/fc1c19172fe78c6a7bb2b4c37a30a339>

3 WORD-Dokumente

Das WORD-Dokument, das die Akronyme enthält, wird im Folgenden *Zieldokument* genannt. Die entsprechende WORD-Datei besitzt gewöhnlich den Zusatz **docx** zum Dateinamen.

Das WORD-Dokument, das die Makros (Prozeduren) enthält, wird im Folgenden *Quelldokument* genannt. Die entsprechende WORD-Datei besitzt gewöhnlich den Dateizusatz **docm**.

Die Prozeduren im *Quelldokument* können immer wieder auf verschiedene *Zieldokumente* angewendet werden, aber sinnvollerweise nur dann, wenn die oben genannten syntaktischen Voraussetzungen (s. Seite 2) erfüllt sind.




Abbildung 1: Zieldokument(e) und Quelldokument

4 Tabellenaufbau

Die Tabelle der Akronyme besteht aus 3 Spalten:

- Akronym
- Definition des Akronyms
- Seitennummer des ersten Vorkommens des Akronyms

Wenn ein Akronym beim automatisierten Suchen im Dokument gefunden wird, werden die erste und dritte Spalte mit den entsprechenden Angaben gefüllt. Wenn zu einem Akronym eine vor- oder nachgestellte Definition gehört, wird diese ebenfalls automatisch gefunden und jeweils in die zweite Tabellenspalte eingesetzt. Sonst wird die zweite Tabellenspalte mit „*nicht gefunden*“ belegt und muss später manuell berichtigt werden.

Wenn mehr als 1 Akronym in Zieldokument gefunden wird, wird die Tabelle nach dem Inhalt in der ersten Spalte aufsteigend sortiert.

Die Spaltenköpfe in der Kopfzeile der Tabelle werden fett formatiert und ihr Hintergrund hellgrau schattiert.

5 Organisation der Prozeduren

Zum Finden und Tabellieren von Akronymen in einem Zieldokument werden diese 4 Makros benutzt (siehe Abbildung 2):




Abbildung 2: Organisation der Prozeduren

Die Hauptaufgabe von **FindListOfAcronyms** ist, im Zieldokument die Überschrift *Abkürzungsverzeichnis* zu finden und danach die Unterprozedur **ListAcronyms** aufzurufen.

Die Unterprozedur **ListAcronyms** sucht im Zieldokument nach Akronymen in einem Abschnitt (engl. *section*), dessen erster Absatz (engl. *paragraph*) mit der eingebauten Formatvorlage *Überschrift 1* verbunden ist. Es wird dabei angenommen, dass nur der zugehörige Abschnitt nach Akronymen durchsucht werden soll, also *nicht* das ganze Zieldokument. Die beiden untergeordneten Prozeduren (siehe Abbildung 2) werden von dort aus automatisch aufgerufen und ausgeführt.

Die Unterprozeduren **DeleteOldTable** und **FormatNewTable** dienen im Wesentlichen dazu, die Wiederholbarkeit der Anwendung sicherzustellen. Diese Eigenschaft ist wichtig, wenn im Zieldokument nachträglich Änderungen vorgenommen werden, die zu einer Verschiebung der Seitenzahlen führen. Dann müssen die Seitenzahlen der Akronyme aktualisiert werden.

Wie erwähnt, befinden sich die Prozeduren als sog. *Makros* in einem beliebig benanntem Quelldokument mit dem Namenszusatz ***.docm**. Es enthält gewöhnlich keinen Fließtext, sondern nur die vier genannten Makros. Wie diese in einem Quelldokument installiert werden können, wird sehr anschaulich beschrieben von Graham Mayor (Mayor, 2012). Die erforderlichen Installationsschritte werden deshalb hier nicht weiter beschrieben.

Das WORD-Dokument, in dem sich die gesuchten Akronyme befinden und wo sie in einer Tabelle unter der Überschrift „*Abkürzungsverzeichnis*“ ausgewiesen werden sollen, wird – wie gesagt – Zieldokument genannt. Der Namenszusatz ist gewöhnlich ***.docx**. Der Name des Zieldokuments wird in der ersten Unterprozedur als Konstante (conTgtNm) hinterlegt.

5.1 Abkürzungsverzeichnis finden

Hauptaufgabe des Programms *FindListOfAcronyms* ist, genau die Stelle im Zieldokument zu finden, wo die Tabelle der Akronyme eingefügt werden soll. Nur wenn diese Stelle gefunden wird, kann dort die genannte Tabelle erstellt werden.

5.2 Tabelle einfügen

Die Unterprozedur *ListAcronyms* wird vom Makro *FindListOfAcronyms* aufgerufen und sorgt dafür, dass die Tabelle der Akronyme an vorgegebener Stelle im Zieldokument eingesetzt wird, nämlich bei der Textmarke **TabSpot** unterhalb der Überschrift „*Abkürzungsverzeichnis*“.

Zunächst wird die möglicherweise bereits vorhandene Tabelle der Akronyme gelöscht, um Platz für eine neue, aktuelle Tabelle zu schaffen. Dadurch ist es möglich, den gesamten Suchvorgang beliebig oft zu wiederholen. Das ist vorteilhaft bei umfangreichen Dokumenten, wenn sich die Seitennummern durch nachträgliche Änderungen verschieben.

5.3 Suchbereich bestimmen

Das Programm *ListAcronyms* bestimmt u. a., in welchem Abschnitt (engl. section) des Zieldokuments nach Akronymen gesucht werden soll. Dieser Suchbereich wird festgelegt, indem nach dem ersten Vorkommen eines Absatzes (engl. paragraph) gesucht wird, der mit der in WORD eingebauten Formatvorlage „*Überschrift 1*“ formatiert ist. Wegen der vorausgesetzten Einteilung des Zieldokuments nach logischen Abschnitten beginnt genau dort der Abschnitt, der durchsucht werden soll. Damit wird erreicht, dass nicht relevante Abschnitte des Zieldokuments bei der Suche unberücksichtigt bleiben.

5.4 Suchbereich durchlaufen

Mit einer *Do-Until-Schleife* wird im zuvor festgelegten Suchbereich nach Akronymen gesucht. Das dafür verwendete Suchmuster, gespeichert in der Variablen *strPattern*, besteht aus einem sog. *regulären Ausdruck*⁶:

```
strPattern = "<[A-Z]{2" & strListSep & ">"
```

⁶ Mit einem *regulären Ausdruck* (engl. regular expression, kurz: regex oder regexp) werden Zeichenketten auf bestimmte Eigenschaften bzw. Muster geprüft.

Dieses Suchmuster bedeutet von links nach rechts:

- < Beginn des gesuchten Wortes
- [A-Z] Zeichenauswahl in eckigen Klammern: Hier nur Großbuchstaben von A bis Z
- { Beginn der Wiederholungsfaktoren, sog. Quantoren
- 2 zwei oder mehr Zeichen aus der Zeichenauswahl [A-Z]
- & Operator zur Verkettung von Zeichenketten
- `strListSep` Die Variable `strListSep` enthält das jeweils gültige Trennzeichen⁷ für Listen
- & Operator zur Verkettung von Zeichenketten
- } Ende der Wiederholungsfaktoren
- > Ende des gesuchten Wortes

Mit dem Suchmuster in `strPattern` werden Zeichenketten mit 2 oder mehr Großbuchstaben in Folge als gesuchte Akronyme identifiziert. Zusammen mit der jeweils dazu gehörigen Seitennummer werden diese beiden Informationen zwischengespeichert, um daraus in einem späteren Schritt die nach Akronymen geordnete Tabelle der gefundenen Akronyme zu erzeugen.

5.5 Suchergebnisse ausgeben

Die zwischengespeicherten Informationen werden an der gewünschten Stelle hinter der Überschrift „*Abkürzungsverzeichnis*“ gefügt. In einem nächsten Schritt werden sie in eine dreispaltige Tabelle umgewandelt⁸. Wird mehr als 1 Akronym gefunden, wird die Tabelle nach dem Inhalt in der 1. Spalte aufsteigend sortiert. In einem letzten Schritt wird das Aussehen der Tabelle verschönert. Dazu dient die Prozedur mit dem Namen `FormatNewTable`.

5.6 Umgebenden Text finden

Wenn im definierten Suchbereich nicht nur Akronyme gefunden werden sollen, sondern auch die jeweils dazugehörige Definition, müssen die Codezeilen aus Listing 5 an der grau hinterlegten Stelle in Listing 3 eingefügt werden. Lediglich aus didaktischen Gründen werden diese Codezeilen in Listing 5 separat dargestellt. Es ist aber praktischer, auf die Aussonderung zu verzichten.

Je nach Schreibweise der gefundenen Akronyme mit vor- oder nachgestellter Definition wird links bzw. rechts davon nach der Definition (d. h. Bedeutung) gesucht, siehe Listing 5.

Wird ein Akronym im relevanten Abschnitt des Zieldokuments gefunden aber nicht die zugehörige Definition, wird der Hinweis (*nicht gefunden*) in die entsprechende Zelle der Akronym-Tabelle eingefügt. Die nicht gefundene Definition muss später manuell berichtigt werden.

⁷ Je nach Land entweder ein Komma oder ein Semikolon (letzteres für Deutschland)

⁸ Das geht erheblich schneller, als wenn eine bestehende Tabelle Zeile für Zeile und Zelle für Zelle belegt wird.

6 Funktionstest

Die Eignung der vorgestellten Unterprozeduren zur Erstellung einer dreispaltigen Tabelle mit Akronymen nach der Überschrift „*Abkürzungsverzeichnis*“ soll mit 3 Kurzwörtern aus dem Bereich der Automarken demonstriert werden:

- VW (Volkswagen)
- (Bayerische Motorenwerke) BMW
- AUDI

Das Ergebnis des Funktionstests ist im vorliegenden Dokument unter der Überschrift „*Abkürzungsverzeichnis*“ zu sehen (vgl. Seite V). Wie erwartet werden 3 Einträge für die soeben erwähnten Kurzwörter vorkommen.

Die Definition des Kurzworts AUDI kommt im Suchbereich *nicht* vor und folglich auch nicht in der Tabelle. Der fehlende Eintrag in der zweiten Spalte der Akronym-Tabelle muss daher manuell nachgetragen werden.

Die Seitennummern in der dritten Spalte der Tabelle geben das erste Vorkommen eines Akronyms an. Jedes weitere Vorkommen wird ignoriert, vgl. dazu in Listing 3 folgende If-Abfrage:

```
If Instr(strOutput, strAcronym) = 0 Then ... End If
```

Die VBA-Textfunktion *Instr* gibt 0 zurück, wenn *strAcronym* innerhalb von *strOutput* nicht vorkommt.

7 Zusammenfassung

Die Besonderheiten des beschriebenen Ansatzes zum Tabellieren von Akronymen im Abkürzungsverzeichnis eines Zieldokuments sind:

- Einsatz strukturierter, ausführlich kommentierter VBA-Prozeduren, die sich im WORD-Quelldokument befinden.
- Möglichkeit zur Wiederverwendung dieser Prozeduren für beliebig viele Zieldokumente, die jeweils den beschriebenen *Voraussetzungen* (s. Seite 2) entsprechen.
- Berücksichtigung der Definition von Akronymen bei syntaktisch richtiger Schreibweise im Text des Zieldokuments.
- Eingrenzung der Suche nach Akronymen auf einen genau bestimmbar Abschnitt des Zieldokuments.

Listings der Prozeduren

Die Listings der Prozeduren enthalten viele Kommentare, damit sie auch von Personen verstanden werden können, die sich mit dem Schreiben von VBA-Makros nicht auskennen. Die Kommentare dienen im Wesentlichen dazu, die Abläufe der jeweiligen Prozedur zu verdeutlichen.

```
Sub FindListOfAcronyms()  
  Const conTgtNm As String = "Automarken" ' Name d.Zieldokuments: Bitte hier abpassen!  
  Const conExt As String = ".docx" ' Dateizusatz  
  Dim docSrc As Document ' Quelldokument  
  Dim docTgt As Document ' Zieldokument  
  Dim strTgtPath As String ' Pfad zum Zieldokument  
  Dim strTgtFullNm As String ' voller Name des Zieldokuments  
  Dim lngSctn As Long ' Nr. d. relevanten Abschnitts in docTgt  
  Application.ScreenUpdating = False ' Bildschirmaktualisierung ausschalten  
  On Error GoTo ErrHandler ' Fehlerbehandlungsroutine  
  Set docSrc = ActiveDocument ' Verweis zum Quelldokument setzen  
  strTgtPath = docSrc.Path ' Pfad zum Zieldokument bestimmen  
  If Right(strTgtPath, 1) <> "\" Then strTgtPath = strTgtPath + "\"  
  strTgtFullNm = strTgtPath & conTgtNm & conExt  
  If Dir(strTgtFullNm) = vbNullString Then  
    MsgBox "Word-Datei '" & strTgtFullNm & "' nicht gefunden!", vbCritical  
    GoTo ExitHandler  
  End If  
  Documents.Open FileName:=strTgtFullNm ' Zieldatei öffnen  
  Set docTgt = ActiveDocument ' Verweis zum Zieldokument setzen  
  docTgt.Select ' Zieldokument auswählen  
  With Selection.Find ' Überschrift 'Abkürzungsverzeichnis' finden  
    .ClearFormatting  
    .Forward = True  
    .Format = True  
    .Style = "ÜberschriftOhneNr" ' benutzerdefinierte Formatvorlage  
    .MatchCase = True  
    .MatchWholeWord = True  
    .Wrap = wdFindStop  
    .Execute FindText:="Abkürzungsverzeichnis"  
  End With  
  If .Found = True Then ' Wenn Suche erfolgreich, dann ...  
    Selection.Collapse wdCollapseEnd ' Ergebnis d. Suche auf d. Endpunkt reduzieren  
    Call ListAcronyms(docTgt) ' Tabelle d. Akronyme im Zieldokument erstellen  
  End If  
ExitHandler:  
  On Error Resume Next  
  If docTgt.Saved = False Then docTgt.Save ' Zieldokument speichern, aber nicht schließen  
  Set docTgt = Nothing ' Verbindung zum Zieldokument aufheben  
  Application.VBE.MainWindow.Visible = False ' VBA-Editor ausblenden  
  docSrc.Close SaveChanges:=wdSaveChanges ' Quelldokument schliessen und speichern  
  Set docSrc = Nothing ' Verbindung zum Quelldokument aufheben  
  Application.ScreenUpdating = True ' Bildschirmaktualisierung einschalten  
  Exit Sub ' Prozedur verlassen  
ErrHandler:  
  MsgBox Err.Description, vbCritical ' Fehlermeldung anzeigen  
  Resume ExitHandler ' Programm bei der Zeilenmarke "ExitHandler" fortsetzen  
End Sub
```

Listing 1: Das Wort *Abkürzungsverzeichnis* im Zieldokument finden

Die benutzerdefinierte Formatvorlage *ÜberschriftOhneNr* legt u.a. fest, dass ein *Seitenumbruch oberhalb* des *Abkürzungsverzeichnisses* vorgenommen wird. Im Programm (siehe Listing 1) wird die Suche nach dem Wort *Abkürzungsverzeichnis* dadurch eingeschränkt.

```

Sub ListAcronyms(docTgt As Document)
' Aufgabe: Gefundene Akronyme tabellieren.
' Die Konstante conBkmNm enthält den Namen der Textmarke, die bestimmt,
' wo die Tabelle der Akronyme eingefügt werden soll.
Const conBkmNm As String = "TabSpot"
Dim bolEmpty As Boolean ' Indikator für leere Absätze
Dim intCnt As Integer ' Zähler für gefundene Akronyme
Dim lngAsk As Long ' Antwort auf Abfrage
Dim lngSctn As Long ' Abschnittsnummer
Dim lngPgNbr As Long ' Seitennummer eines Akronyms
Dim newTbl As Table ' neue Tabelle für gefundene Akronyme
Dim para As Paragraph ' aktueller Absatz
Dim rngTbl As Range ' Tabellenbereich
Dim rngSctn As Range ' Suchbereich für Akronyme
Dim rngResult As Range ' verkleinerter Suchbereich für Akronyme
Dim rngPara As Range ' Absatzbereich
Dim strAcronym As String ' Akronym (d. h. Kurzwort, Kürzel)
Dim strOutput As String ' Zeichenkette als vorläufige Tabellenzeile
Dim strDef As String ' Definition/Erklärung/Bedeutung eines Akronyms
Dim strListSep As String ' Trennzeichen f. Listen
Dim strPattern As String ' Suchmuster (regulärer Ausdruck)
On Error GoTo ErrHandler ' Fehleroutine definieren
With Application
.ScreenUpdating = False ' Bildschirmaktualisierung ausschalten
.strListSep = .International(wdListSeparator) ' Trennzeichen f. Listen ermitteln
End With
' Anzeige von ausgeblendetem Text verhindern
ActiveWindow.View.ShowHiddenText = False
' Alte Tabelle der Akronyme löschen, falls die zugehörige Textmarke vorhanden ist
Call DeleteOldTable(docTgt, conBkmNm)
' Ersten Absatz mit der Formatvorlage 'Überschrift 1' im Zieldokument finden
For Each para In docTgt.Paragraphs
If para.Style.NameLocal = "Überschrift 1" Then
Set rngPara = para.Range
rngPara.Select
' relevante Abschnittsnummer im Zieldokument bestimmen
lngSctn = rngPara.Information(wdActiveEndSectionNumber)
Exit For
End If
Next para
If lngSctn > 0 Then ' Wenn die relevante Abschnittsnummer im Zieldokument gefunden wurde, ...
' Such- und Ergebnisbereich für Akronyme im Zieldokument festlegen
Set rngSctn = docTgt.Sections(lngSctn).Range
Set rngResult = rngSctn.Duplicate
Else
MsgBox "Abbruch: Relevanten Abschnitt im Zieldokument nicht gefunden", vbCritical
GoTo ExitPoint ' Prozedur abbrechen
End If

```

Listing 2: Gefundene Akronyme tabellieren, Teil 1 von 3

Die grau markierte *For-Next*-Schleife in Listing 2 dient dazu, genau den Abschnitt (engl. *section*) im Zieldokument zu bestimmen, in dem nach Akronymen gesucht werden soll.

```

' Suchmuster (regulärer Ausdruck) definieren: Findet 2 oder mehr Großbuchstaben in Folge
strPattern = "<[A-Z]{2" & strListSep & ">"
' Kopfzeile der (neuen) Tabelle für Akronyme definieren
strOutput = "Akronym" & vbTab & "Definition" & vbTab & "Seite" & vbCr
Do ' fußgesteuerte Suchschleife
  With rngResult.Find
    .ClearFormatting
    .Text = strPattern          ' Suchmuster für Akronyme
    .Replacement.Text = vbNullString
    .Forward = True
    .Wrap = wdFindStop
    .Format = False
    .MatchCase = True
    .MatchWildcards = True    ' Platzhaltersuche
    .MatchWholeWord = True
    .Execute
  End With
  If Not rngResult.Find.Found Then Exit Do
  rngResult.Select
  With Selection
    strAcronym = .Text          ' Akronym
    lngPgNbr = .Information(wdActiveEndAdjustedPageNumber) ' Seitennummer
  End With

'<---Codezeilen in Listing 5 ggf. hier einfügen bei bestimmter Schreibweise der Akronyme --->

' Wenn das gefundene Akronym noch nicht im Output-String erhalten ist, dann ...
If InStr(strOutput, strAcronym) = 0 Then
  intCnt = intCnt + 1 ' Zähler für Akronyme erhöhen
  ' Eine weitere neue Tabellenzeile vorbereiten
  strOutput = strOutput & strAcronym & _
    vbTab & strExplain & _
    vbTab & Str(lngPgNbr) & vbCr
End If
' Suchbereich im relevanten Abschnitt verkleinern
rngResult.MoveStart unit:=wdCharacter, Count:=Len(strAcronym)
rngResult.End = rngSctn.End
Loop Until Not rngResult.Find.Found

```

Listing 3: Gefundene Akronyme tabellieren, Teil 2 von 3

Die grau markierte *Do-Until*-Schleife (siehe Listing 3) dient dazu, im zuvor ausgewählten Abschnitt des Zieldokuments nach Akronymen zu suchen und diese in der Zeichenkette *strOutput* zu sammeln.

Die grau markierte Kommentarzeile im Schleifenkörper dient als Platzhalter für Codezeilen aus Listing 5, die genau hier eingebaut werden können.

```

If intCnt = 0 Then
    MsgBox "Keine Akronyme im relevanten Suchbereich gefunden", vbExclamation
    GoTo ExitPoint ' Prozedur abbrechen
End If
' Vorbereitete Tabellenzeilen hinter "Abkürzungsverzeichnis" einfügen
docTgt.Bookmarks(conBkmNm).Select
With Selection
    .TypeParagraph
    .InsertBefore strOutput
End With
' Vorbereitete Tabellenzeilen in eine neue Tabelle mit 3 Spalten umwandeln
Set newTbl = Selection.ConvertToTable( _
    Separator:=wdSeparateByTabs, NumRows:=Selection.Paragraphs.Count, NumColumns:=3)
' Neue Tabelle der Akronyme ggf. sortieren
With newTbl
    If .Rows.Count > 1 Then
        .Sort ExcludeHeader:=True, FieldNumber:="Column 1", _
            SortFieldType:=wdSortFieldAlphanumeric, _
            SortOrder:=wdSortOrderAscending
    End If
End With
Call FormatNewTable(newTbl) ' Neue Tabelle der Akronyme formatieren
Set rngTbl = newTbl.Range
' Leeren Absatz VOR der Akronym-Tabelle ggf. löschen
With rngTbl
    .Collapse Direction:=wdCollapseStart
    .Move unit:=wdParagraph, Count:=-1
    With .Paragraphs(1).Range
        .Select
        If Not Selection.Information(wdWithInTable) Then
            .HighlightColorIndex = wdYellow
            If .Text = vbCr Then
                lngAsk = MsgBox("Leeren Absatz VOR der Akronym-Tabelle entfernen?", _
                    vbYesNoCancel + vbQuestion, "Leerer Absatz")
                If lngAsk = vbYes Then
                    .Delete
                ElseIf lngAsk = vbCancel Then
                    GoTo ExitPoint
                End If
            End If
        End If
    End With
End With
' Leeren Absatz NACH der Akronym-Tabelle ggf. löschen
rngTbl.Collapse Direction:=wdCollapseEnd
With rngTbl.Paragraphs(1).Range
    .Select
    If Not Selection.Information(wdWithInTable) Then
        .HighlightColorIndex = wdYellow
        If .Text = vbCr Then
            lngAsk = MsgBox("Leeren Absatz NACH der Akronym-Tabelle entfernen?", _
                vbYesNoCancel + vbQuestion, "Leerer Absatz")
            If lngAsk = vbYes Then
                .Delete
            ElseIf lngAsk = vbCancel Then
                GoTo ExitPoint
            End If
        End If
    End If
End With
ExitPoint:
On Error Resume Next
Application.ScreenUpdating = True ' Bildschirmaktualisierung einschalten
Selection.HomeKey unit:=wdStory
MsgBox "Fertig: " & Str(intCnt) & " Akronyme gefunden." & vbCrLf & _
    "Tabelle der Akronyme in das Abkürzungsverzeichnis des Zieldokuments " & vbCrLf & _
    docTgt.FullName & vbCrLf & "eingefügt.", vbInformation
Set rngTbl = Nothing: Set rngPara = Nothing: Set rngSctn = Nothing: Set rngResult = Nothing
Exit Sub
ErrorHandler:
MsgBox "Fehler " & Err.Description, vbCritical
Resume ExitPoint
End Sub

```

Listing 4: Gefundene Akronyme tabellieren, Teil 3 von 3

Die in der Zeichenkette *strOutput* gesammelten Informationen (s. Listing 3, Teil 2) werden an definierter Stelle hinter der Überschrift *Abkürzungsverzeichnis* im Zieldokument in eine WORD-Tabelle umgewandelt. Diese Tabelle wird mit der Unterprozedur *FormatNewTable* aufbereitet. Anschließend wird aufgeräumt.

```
' Bestimmte Schreibweise der Akronyme
strDef = vbNullString ' Anfangswert für die Definition eines Akronyms
' Schreibweise im Text: ACRONYM (Definition)
With Selection
    .MoveRight Unit:=wdWord
    .MoveRight Unit:=wdCharacter, Extend:=wdExtend
    If .Text = "(" Then
        Do While .Text <> ")"
            strDef = strDef & .Text
            .Collapse Direction:=wdCollapseEnd
            .MoveRight Unit:=wdCharacter, Extend:=wdExtend
        Loop
    End If
    .Collapse Direction:=wdCollapseEnd
    If Left(strDef, 1) = "(" Then
        strDef = Mid(strDef, 2, Len(strDef))
    End If
    ' Schreibweise im Text: (Definition) ACRONYM
    If Len(strDef) = 0 Then ' Wenn Definition noch nicht gefunden wurde, dann ...
        .MoveLeft Unit:=wdCharacter, Count:=1, Extend:=wdMove
        .MoveLeft Unit:=wdWord
        .MoveLeft Unit:=wdCharacter, Count:=2, Extend:=wdMove
        If .Text = ")" Then
            Do While .Text <> "("
                strDef = strDef & .Text
                .Collapse Direction:=wdCollapseStart
                .MoveLeft Unit:=wdCharacter, Extend:=wdExtend
            Loop
        End If
        .Collapse Direction:=wdCollapseEnd
        If Left(strDef, 1) = ")" Then
            strDef = StrReverse(Mid(strDef, 2, Len(strDef)))
        End If
    End If
End With
If strDef = vbNullstrig Then strDef = "(nicht gefunden)"
End If
```

Listing 5: Definition der Akronyme ermitteln bei bestimmter Schreibweise im Text⁹

Die in Listing 5 enthaltenen Codezeilen sind optional. Sie können an der grau markierten Stelle in Listing 3 eingefügt werden.

⁹ Die Codezeilen für die Schreibweise Akronym (Definition) wurden erstellt in Anlehnung an (Wyatt, 2015).

```

Sub DeleteOldTable(docTgt As Document, strBkm As String)
' Alte Tabelle im Zieldokument (docTgt) löschen
' falls die zugehörige Textmarke (strBkm) existiert,
' sonst die Textmarke erneuern.
With docTgt.Bookmarks
  If .Exists(strBkm) Then ' Wenn Textmarke existiert, ...
    Selection.GoTo What:=wdGoToBookmark, Name:=strBkm
    If Selection.Information(wdWithInTable) Then
      Selection.Tables(1).Delete ' alte Tabelle löschen
    End If
  Else ' Textmarke für Tabelle der Akronyme erneuern
    Selection.TypeParagraph
    .Add Name:=strBkm, Range:=Selection.Range
    .DefaultSorting = wdSortByName
    .ShowHidden = False
  End If
End With
End Sub

```

Listing 6: Alte Tabelle der Akronyme löschen und zugehörige Textmarke erneuern

Die Prozedur *DeleteOldTable* ist wichtig für die Wiederholbarkeit der Suche nach Akronymen im Zieldokument.

```

Sub FormatNewTable(tblNew As Table)
Dim objCell As Cell
Dim lngRow As Long
With tblNew
  .Rows.Alignment = wdAlignRowCenter ' Tabellenzeilen zentrieren
  If .Style <> "Tabellenraster" Then
    .Style = "Tabellenraster"
  End If
  With .Rows(1)
    .HeadingFormat = True ' als Kopfzeile formatieren
    With .Range
      .Shading.BackgroundPatternColor = wdColorGray25 ' Kopfzeile hellgrau schattieren
      .Font.Bold = True ' Kopfzeile fett drucken
    End With
  End With
  For lngRow = 2 To .Rows.Count
    For Each objCell In .Rows(lngRow).Cells
      objCell.Range.Font.Size = 10 ' Schriftgrad auf 10 Punkte festlegen
    Next objCell
  Next lngRow
  ' Seitennummern in der letzten Tabellenspalte horizontal und vertikal zentrieren
  For Each objCell In .Columns(.Columns.Count).Cells
    With objCell
      .Range.ParagraphFormat.Alignment = wdAlignParagraphCenter
      .VerticalAlignment = wdCellAlignVerticalCenter
    End With
  Next objCell
  ' Breite der drei Tabellenspalten prozentual festlegen
  .Columns.PreferredWidthType = wdPreferredWidthPercent
  .Columns(1).PreferredWidth = 25
  .Columns(2).PreferredWidth = 60
  .Columns(3).PreferredWidth = 15
End With
End Sub

```

Listing 7: Neue Tabelle der Akronyme formatieren

Die Unterprozedur *FormatNewTable* führt die in Listing 4, Teil 3, gelb markierte Aufgabe durch.

Literatur-/Quellenverzeichnis

Fredborg, L. (2008). *Extract ACRONYMS to New Document*. Abgerufen am 10. 7 2016 von http://www.thedoctools.com/downloads/basACRONYMS_Extract.htm

Mayor, G. (2012). *Idiot's guide to installing macros*. Abgerufen am 29. 6 2016 von Graham Mayor - Word Pages: http://www.gmayor.com/installing_macro.htm

o.V. (21. 6 2016). *Akronym*. Abgerufen am 2. 7 2016 von WIKIPEDIA:
<https://de.wikipedia.org/wiki/Akronym>

Puls, Erika. (13. 6 2013). *Wissenschaftliche Arbeit #1 mit Word 2010 formatieren*. (HAWK)
Abgerufen am 29. 6 2016 von <https://www.youtube.com/watch?v=e53GMQpZAzE>

Wyatt, A. (22. 6 2015). *Auto Creation of an Acronym List*. Abgerufen am 2. 7 2016 von Allen Wyatt's WordTips:
http://wordribbon.tips.net/T010935_Auto_Creation_of_an_Acronym_List.html